
Part I: Instance-Based Components 
Thus far, most of the components we have looked at have been static components, generally 

instantiated on use and has no associated data. Instance-based components, however, do 

have data associated with them and they are usually instantiated in a persistent memory 

scope.  You can then use properties to access persistent data, through either publicly 

accessible properties, which are read directly from any ColdFusion page using dot-notation, or 

privately accessible properties that are only available through calling a CFC method.  

 

So why use instance-based components over "normal" static components? Instance-based 

components allow for better encapsulation of code, significantly faster execution times, and 

allows for development using object-oriented design patterns. There are some caveats, of 

course. When you instantiate a CFC into a shared-scope, the methods within it will also be 

cached within that scope. So, if you modify your method, you will have to clear the shared 

scope before those changes are implemented. Also, using this method, you cannot 

consistently access any of the variable scopes (for, application, session, etc), so you need to 

follow function "best practices" and pass in every variable manipulated/used within the function 

as an argument. 

Instantiating 

Instance-based components are typically instantiated using either the <cfobject> tag or the 

CreateObject() function. This is used to place an instantiated, data-bound component into a 

shared scope. When the component is first instantiated, you typically invoke an init() method to 

pass in arguments that you want to persist for that component instance. The CreateObject() 

tends to be the more flexible way of doing this, and allows for generally writing only one line of 

code instead of two. 

<!-- Using cfobject --> 

<cfobject name="cfcSpecials" component = 
#application.componentpath#Specials"> 

<cfset cfcSpecials = cfcSpecials.init(application.datasource)> 

 

<!-- Using CreateObject() --> 

<cfset cfcSpecials = CreateObject("component", 
"#application.componentpath#Specials").init( datasource = 
application.datasource)> 

When you persist a component in a shared scope, you also need to keep in mind the same 

locking considerations you would have for any shared scope variable. Namely, you must 

decide whether there may be any data integrity problems for concurrent access. If there are, 

you either need to place locks around the instantiations of the object methods, or place locks 

within the CFC methods themselves.  The latter is the best practice to follow, particularly in a 

multi-developer environment as it doesn't require communication with developers about when 

and if locks need to be set in the display tier or within dependent CFCs. 

 

In addition to persisting component instances, you can persist data inside of an instance. 

Using this methodology, the component itself becomes its own variable scope, analogous to 

the session, application, and variables scopes. When you persist data in this fashion, the data 

is considered a property of the component. 

 

Properties can be public or private, based on how they are set and accessed. Public properties 

are created using the THIS scope within a component and may be accessed from anywhere 

using object.property dot-syntax. They can also be documented using the <cfproperty> tag, as 

we will see shortly. Private properties are global to all methods within the component, but 

hidden from any other ColdFusion file. They can be accessed through what is referred to as 

getter/setter methods, a series of user-defined functions for modifying and returning their 

values, but can not be documented through the <cfproperty> tag nor any of the built-in CFC 

self-documenting features. 

Public properties 

You create public properties the same as you would any other variable, through <cfset> or 

<cfparam>, except you must use the THIS prefix. For example, if you place the following code 

directly after the <cfcomponent> tag: 

<cfset THIS.datasource="CF8advOnYourOwn"> 

The data source variable will now be available to all methods in the component and be 

available for the life of component. You can create these variables inside any method, however 

for most variables you will want to initialize them directly after the <cfcomponent> method so 

they are initialized when the component instance is instantiated. This makes them instantly 

and always available for use and updating by the methods. In this example, we initialize a data 

structure that mimics the database table the component functions will later operate against: 

<cfcomponent> 

<cfset this.datasource = "CF8advOnYourOwn"> 

<cfset this.instance = { 

administratorid = "", 

username = "", 

password = "", 

administratorRole = "" 

}> 

<cffunction name="Load" output="false" access="public" 
returntype="boolean"> 

<cfargument name="administratorid" type="numeric" required="yes"> 

<cfset var q = ""> 

<cfquery name="q" datasource="#this.datasource#"> 

SELECT * 

FROM administrator 

WHERE administratorid = 

<cfqueryparam cfsqltype="cf_sql_numeric" 
value="#arguments.administratorid#"> 

</cfquery> 

<cfset this.instance = { 

administratorid = q.administratorid, 

username = q.username, 

password = q.password, 

administratorRole = q.administratorRole 

}> 

<cfreturn true> 

</cffunction> 

</cfcomponent> 

 

The instance structure in this example is created only when the CFC is first instantiated. 

Calling the load() function will then populate it with the appropriate values using the passed in 

argument of administratorid.  

<cfset session.cfcAdministrator = createObject("component", 
"#application.cfcpath#Administrator" )> 

<cfset session.cfcAdministrator.Load(2)> 

 

You can reference CFC properties within a component using the THIS prefix, same as when 

you set them. They can be manipulated however you desire: create, change or delete. Public 

properties can also be accessed outside the component using the object.property notation, 

similar to how you can use the object.method notation to reference a component's method. 

Referring again to our example above, we can pull the name of the datasource we are using: 

<cfoutput> 

#session.cfcAdministrator.DataSource# 

</cfoutput> 

These public properties are both readable and writable. Once our component has been 

instantiated, we could reset the datasource property within our code: 

<cfset session.cfcAdministrator.DataSource="CF8advStepByStep"> 

Documenting With CFPROPERTY 

We've already discussed how you can view and augment the self-documentation that is 

generated by ColdFusion for a component. However, if you rely only on the self-

documentation, when you attempt to view the documentation on a component that is using 

public properties, you'll find that the properties field is still blank!   

 
Figure 1: Self-documentation in the Administrator.cfc component that contains public properties but has 

no documentation on them 

To display information on the component properties, we need to add additional documentation 

using the <cfproperty> tag. You use one tag for each variable from the this scope that you 

want to document as an "official" property of the component, with all <cfproperty> tags being 

placed directly after the <cfcomponent> tag and BEFORE you begin initializing the variables. 

<cfcomponent hint="Administrator Accounts for the CMS"> 

<cfproperty name="datasource" hint="Datasource to retrieve info" 
type="string"> 

<cfproperty name="instance.administratorid" hint="Primary key for 
account" type="numeric"> 

<cfproperty name="instance.username" hint="Login name for CMS" 
type="string"> 

<cfproperty name="instance.password" hint="Password for user account" 
type="string"> 

<cfproperty name="instance.administratorRole" hint="Role(s) for roles-
based sec" type="string"> 

<cfset this.datasource="CF8advOnYourOwn"> 

The <cfproperty> tag only has one required attribute, the "name" attribute which is the variable 

name (sans the THIS scope). It also offers five optional attributes. 

 Default - the initial value of the property 

 DisplayName - An alternative name for the property for displaying on screen (appears in 

parenthesis behind the property name) 

 Hint - A text description of the property 

 Required - yes/no as to whether the property is required; must be set to no or 

undeclared if using a default value 

 Type - the data type of the property: any, array, binary, boolean, date, guid, numeric, 

query, string, struct, uuid, variableName, or a component name; if no type is passed, it 

defaults to "any" 

<cfproperty  

    name="name" 

    default="default value" 

    displayname="descriptive name" 

    hint="extended description" 

    required="boolean" 

    type="type"> 

If we set the <cfproperty> information in the component we saw earlier, the component 

documentation would now provide the appropriate details: 

Figure 2: The Administrator component, with <cfproperty> used to document its public properties 

If you properly use the <cfproperty> tags to set property information, you can also view the 

property information directly within Dreamweaver and ColdFusion Builder: 

 
Figure 3: Viewing the CFC properties within Dreamweaver 

Figure 4: Same CFC in ColdFusion Builder; notice ColdFusion builder does not show the property types 

Private properties and the Getter/Setter approach 

Exposing the global variables in a CFC as public properties leaves them easily modifiable 

which could be dangerous as there may be dependencies between properties where if one 

changes, another should change as well. For example, if the datasource changes, we need to 

trigger a change to other data to re-cache it. While you can cheat and not use the THIS scope, 

this is a debated practice and really should not be followed.  

 

The only proper way to assure that modifying persistent data will trigger an event is by only 

allowing changes to be made through the invocation of a CFC-based function. This approach, 

known as the getter/setter method, involves having two methods for each variable, one to 

retrieve it (getter) and one to change it (setter). Components using this method should also use 

an init() function for use in initializing global variables as keys within a memory structure 

named "instance". 

 

Let's look at a template that you may want to consider using whenever you make a new CFC: 

<cfcomponent name="Administrator"> 

<cfset instance = structnew()> 

<cffunction name="init" access="public" returntype="Administrator" 
output="false"> 

<cfargument name="datasource" required="yes" type="String" 
hint="Data source"> 

<!--- other args omitted for brevity ---> 

<cfset Datasource_Set(arguments.datasource)> 

<cfreturn this> 

</cffunction> 

<cffunction name="DataSource_Get" access="public" returntype="string" 
output="false"> 

<cfreturn instance.datasource> 

</cffunction> 

<cffunction name="Datasource_Set" access="public" returntype="boolean" 
output="false"> 

<cfargument name="datasource" type="string" 

required="yes"> 

<cfset instance.datasource = arguments.datasource> 

<cfreturn true> 

</cffunction> 

</cfcomponent> 

 

Notice that here, our init() method is not using one of the standard returntypes we have been 

working with before. Instead, it is using a pointer to an instantiated component in memory, 

specifically its owning component Administrator, its CFC methods, and local component 

variables (instance) by using just the component name. We also have our first getter and setter 

methods, one for getting the value of the datasource and another to change it. 

 

When we invoke the CFC using CreateObject, we can load in our init variables as well: 

<cfset session.cfcAdministrator = CreateObject("component", 
"Administrator").init("CF8advOnYourOwn")> 

Restricting Invocation of Methods 

The <cffunction> tag that we use for creating functions has an attribute, "access", which 

determines how that function can be shared and who can access it. It accepts four possible 

choices in descending order of restriction: 

Access Type Who Can Call This Function 
private Other functions within the same component and any 

components that extend the component in which this function is 
defined 

package Any other component within the same directory in addition to the 
conditions for access="private" 

public Any ColdFusion page or component method located on the 
same server; can also be called via Flash remoting, unless 
disabled through the CF Administrator 

remote Any ColdFusion page or component method located on any 
server, by Flash (using Flash Remoting MX), or by any platform 
or program as a web service 

Always set the access method to the highest value possible that still allows your application to 

run. Only use remote when necessary to provide a web service or for Flash/Flex/AIR 

applications. 

 

Usage of a function can also be restricted by user roles that have been defined 

programmatically using the <cfloginuser> tag using the "roles" attribute of the <cffunction> tag. 

In the example below, we restrict access to a function, which hard deletes data, to users who 

are logged in with an administrative role. 

<cfcomponent> 

<cffunction name="EmptyTrash" access="public" roles="admin" 
returntype="boolean"> 

<cfquery datasource="#instance.datasource#"> 

DELETE 

FROM article 

WHERE endtime is not null 

</cfquery> 

<cfreturn true> 

</cffunction> 

</cfcomponent> 

OnMissingMethod() 

The OnMissingMethod() is a special generic method you can add to any function that allows 

you to gracefully handle any requests for a non-existent method within your CFC. It has two 

built in arguments: missingmethodname, which contains the name of the method that was 

called, and missingmethodarguments, which is a structure containing any arguments passed 

to the method. 

<cffunction name="onMissingMethod" returntype="any"> 

<cfargument name="missingmethodname" type="string"> 

<cfargument name="missingmethodarguments" type="struct"> 

<!--- business logic goes here ---> 

</cffunction> 

In addition to many other basic applications and more graceful error handling, this function 

makes it easier to implement the getter/setter methodology by allowing you to create a 

dynamic proxy for such calls. This method will not, however, be triggered by a call to a missing 

method that originates from within the same component. 

If we were to continue building out our example component using getter/setter methodology, 

we would have to build one getter and one setter method for each database column. With 

seven columns in our sample table, we'd have fourteen separate functions to include to 

support the framework! Can you imagine doing that for a table with some 30 or 40 fields? 

Obviously, that would be a very bloated, very messy, and hard to manage function! However, 

using the OnMissingMethod() handler, we can implement the same functionality without the 

coding tedium and with far greater flexibility. 

<cffunction name="onMissingMethod" access="public" returntype="any" 
output="true" hint="generalized handler for get/setters"> 

<cfargument name="missingMethodName" type="string"> 

<cfargument name="missingMethodArguments" type="struct"> 

<cfset var action = listlast(arguments.missingmethodname,"_")> 

<cfset var actionvar = listFirst(arguments.missingmethodname,"_")> 

 

<cfswitch expression="#action#"> 

<cfcase value="set"> 

<cfset instance[actionvar] = 
arguments.missingmethodarguments[1])> 

</cfcase> 

<cfcase value="get"> 

<cfreturn instance[actionvar]> 

</cfcase> 

<cfdefaultcase> 

<cfthrow message = "Unknown method invocation: 
#arguments.missingmethodname#"> 

</cfdefaultcase> 

</cfswitch> 

</cffunction> 

Part II: Built-In Components and the Administrator API 
ColdFusion now comes with and automatically recognizes a number of components. In 

ColdFusion 8, this is taken a step further by allowing you to have all components automatically 

be extensions of a central component and by making the ColdFusion Administrator API 

available through a series of components. 

 

Component.cfc is an empty file installed with ColdFusion located in the WEB-INF/cftags 

directory. All components created on the server will automatically inherit any methods and 

properties from this component, making it an effective way to ensure that a set of often-used 

utility functions are available to every component on the server. It can also be used to apply a 

common set of properties to every component, such as a debugging flag. 

 

Most of the options that you can set in the ColdFusion administrator can also be configured 

programmatically by using the built-in Administrator API. This API consists of a number of 

CFCs, corresponding to an area in the ColdFusion administrator, that are located in the 

CFIDE/adminapi directory.  

Component Description 
accessmanager.cfc Specify the username, password, description, access rights, 

sandboxes, and allowed roles for individual users. 

administrator.cfc Contains basic Administrator functionality, including login, 

logout, the Migration Wizard, and the Setup Wizard. You 

must call the login method before calling any other methods 

in the Administrator API. 

base.cfc Base object for all other Administrator API CFCs. 

datasource.cfc Add, modify, and delete ColdFusion data sources. 

debugging.cfc Manage debug settings. 

eventgateway.cfc Manage event gateways. 

extensions.cfc Manage custom tags, mappings, CFXs, applets, CORBA, 

and web services. 

mail.cfc Manage ColdFusion mail settings. 

runtime.cfc Manage runtime settings for fonts, cache, charts, 

configuration, and other settings. 

security.cfc Manage passwords, RDS, and sandbox security. 

serverinstance.cfc Start, stop, and restart JRUN services when running the 

multiserver configuration. 

servermonitoring.cfc Perform many of the Server Monitor tasks programmatically. 

 

All of these CFCs are encrypted, so they cannot be edited nor can the code be reviewed. 

However, you can view the available methods the same as you can any other component, by 

browsing to it in your browser.  If you are using sandbox security, you must enable access to 

the cf_web_root/CFIDE/adminapi directory to use the Administrator API.  

Before using any of the API methods, you must first invoke a login method that passes in your 

ColdFusion Administrator password or the RDS password. In the following example, we use 

the PI to toggle debugging on and off on a server, and adds the current IP to the list of debug 

IPs if it is not already there: 

<cfset cfcAdmin = createObject("component", 
"cfide.adminapi.administrator").login( "password")> 

<cfset cfcDebug = createObject("component", "cfide.adminapi.debugging")> 

<cfset bDebugEnabled = cfcDebug.GetDebugProperty("enableDebug")> 

<cfif not bDebugEnabled>  

<cfset cfcDebug.SetDebugProperty("enableDebug", "True")> 

<!--- add current IP to output list ---> 

<cfif cfcDebug.getIPList() DOES NOT CONTAIN cgi.remote_addr> 

<cfset cfcDebug.setIP(listAppend(cfcDebug.getIPList(), 
cgi.remote_addr))> 

</cfif> 

Debugging is ON 

<cfelse> 

<cfset cfcDebug.SetDebugProperty("enableDebug", "False")> 

Debugging is OFF 

</cfif>  

 

The Administration API is a powerful tool to allow you to build your own custom server 

administrator, ideal for limiting access to various functions on a user-to-user basis. For 

example, one could create a special web application for web applications developers that allow 

them to add needed data sources to the server and clear template caches, while preventing 

them from accessing security and server settings. You could even build an entire custom 

server administrator that allows you to have user-level logging of specific actions to see who 

did what when and that is integrated with other server tools and systems. Obviously, care must 

be taken when using this API to manipulate the ColdFusion server through applications. 

Remember, with great power comes great responsibility, and if you have a system 

administrator, they are unlikely to take kindly to someone abusing the availability of this API.  

Part III: Prototyping Components 
Starting in ColdFusion 8, you can now create pseudo-object oriented definitions for CFC-based 

components. With the <cfinterface> tag, you can created method definitions that must be 

implemented in developer CFCs that reference it. This allows you to validate component 

method and properties to ensure they conform exactly to the original object specification. 

 

For example, a systems architect might create an interface that defines an abstract class of 

Content that is defined as supporting five methods (a “CRUD”): 

• Init() 

• Create() 

• Read() 

• Update() 

• Delete() 

When creating components, he or she instructs the developers to reference this interface for 

file when they create components to support different types of content, perhaps Article.cfc, 

Page.cfc. etc. Interfaces are essentially a shell component, defined as a CFC, but that only 

contains <cffunction> tags and comments, with no business logic at all.  

 

Instead of being wrapped in <cfcomponent> tags, the interface is wrapped in <cfinterface> 

tags. It is considered best practice to preference interface files with the letter “i” or in some 

other similar manner to clearly differentiate them from normal component files. Like 

components, the <cfinterface> tag has an “extends” attribute, allowing it to inherit methods and 

properties from parent interfaces. 

<cfinterface displayname="iContent"> 

<cffunction name="init" access="public"> 

<cfargument name="datasource" type="string"> 

</cffunction> 

<cffunction name="create" hint="creates new instance, stores data in 
database, returns pk" returntype="numeric"> 

<cfargument name="stProperties" type="struct"> 

</cffunction> 

<cffunction name="update" hint="updates instance, stores data in 
database, returns primarykey" returntype="boolean"> 

<cfargument name="pk" type="numeric" hint="primarykey"> 

<cfargument name="stProperties" type="struct"> 

</cffunction> 

<cffunction name="delete" hint="perform soft delete from db" 
returntype="numeric"> 

<cfargument name="pk" type="numeric" hint="Primary key"> 

<cfargument name="updateuser" type="string" hint="Name of user 
requesting deletion"> 

</cffunction> 

<cffunction name="get" hint="Retrieves Data, returns as an array of 
structures" returntype="array"> 

<cfargument name="lpk" type="string" hint="List of primary keys"> 

</cffunction> 

<cffunction name="search" hint="performs a filtered search of the 
content object" returntype="array"> 

<cfargument name="stProperties" type="struct" hint="structure 
containing column/value"> 

</cffunction> 

</cfinterface> 

 

To tell a component to reference an interface, you use the implements attribute of the 

<cfcomponent> tag. It must implement ALL of the methods declared in the referenced 

interface, including being in the same order and having the same arguments (also in the same 

order). A component can implement multiple interfaces by separating each name with a 

comma, however if the interfaces have like named function definitions, those definitions must 

also be identical as ColdFusion does not support function overloading. 

<cfcomponent displayname="Article" implements="iContent"> 

<cffunction name="init" access="public"> 

<cfargument name="datasource" type="string"> 

<cfset instance = structnew()> 

<cfset instance.datasource = arguments.datasource> 

<cfreturn this> 

</cffunction> 

. . . 

</cfcomponent> 

 

The overall usefulness of this tag appears to be still debated, as many feel that ColdFusion’s 

dynamic nature makes it useless. It only is enforced on instantiation and it doesn’t force 100% 

compliance as additional methods can be added without throwing errors. According to Adobe, 

“the interfaces that you define by using this tag can make up the structure of a reusable 

application framework.” Dan Vega described interfaces in this manner: “Interfaces provide a 

contract between the interface and any component that implements it. ColdFusion will make 

sure that contract is fulfilled at run-time.” He also considers them a useful tool to have in one’s 

ColdFusion tool belt that is geared towards more Object Oriented coding methodologies.1 

Primarily, the main usefulness of interfaces appears to be in multi-developer environments 

where developers may be in separate locations, working with separate APIs but that need to 

have the resulting data conform to the same standards for use a central application.2  

 
1 Vega, Dan. cfinterface. Dan Vega: The Ramblings of a ColdFusion addict. July 9, 2007. Retrieved from: 

http://www.danvega.org/blog/index.cfm/2007/7/9/cfInterface on July 16, 2009. 
2 Shuck, Dave. Interfaces in ColdFusion - I think I have finally come around on CFINTERFACE. March 6, 2009. 

Dave Shuck’s InstantSpot.. Retrieved from: http://www.daveshuck.com/blog/2009/03/06/Interfaces-in-
ColdFusion--I-think-I-have-finally-come-around-on-CFINTERFACE on July 16, 2009. 

http://www.danvega.org/blog/index.cfm/2007/7/9/cfInterface
http://www.daveshuck.com/blog/2009/03/06/Interfaces-in-ColdFusion--I-think-I-have-finally-come-around-on-CFINTERFACE
http://www.daveshuck.com/blog/2009/03/06/Interfaces-in-ColdFusion--I-think-I-have-finally-come-around-on-CFINTERFACE

Best Practices and Reminders 
1. Components are the core of advanced development of ColdFusion applications through 

their allowing for a higher degree of code reuse, encapsulation, and security versus 
other CFML constructs 

2. Use instance-based components to better encapsulate your code and significantly 
improve application execution times 

a. Persist instances in the session or application scope for further performance 
improvement 

3. Pass in every variable manipulated or used within a function as an argument, including 
those from shared scopes 

4. When using public properties, use the <cfproperty> tags to properly document any 
public properties of a component and initialize public properties immediately after the 
<cfcomponent> and <cfproperty> tags 

a. Public properties can be read and modified anywhere within your application 
using object.property notation 

b. Use the THIS scope to create and manipulate public properties 

5. Ideally, avoid using the THIS scope and public properties; instead use private properties 
and an init() function to initialize global variables within the instance structure 

a. Use getter/setter methodology to access each variable rather than direct access 

6. Restrict execution of methods using both the access attribute and roles-based security, 
as is appropriate; never set access to remote unless necessary for web services or use 
in Flex or Flash or the like 

7. Use the special method, OnMissingMethod(), to create dynamic getter/setter and to add 
graceful error handling for missing method calls 

8. The special Component.cfc component installed with ColdFusion extends to all 
components on the server automatically, allowing for global methods and properties to 
be distributed to every component instantly 

9. The Administrator API allows for programmatic access to the ColdFusion administrator 
and can be used to create custom ColdFusion server administration systems 

10. Use interfaces to rapidly create a specification for components and their interactions, 
where appropriate. 

Applying What We Learned 
1. Create a new component that has several methods and that uses public properties, 

being sure to document their existence correctly 
a. View your new component's documentation in the browser and your favorite 

editor 
b. Instantiate the component as an instance-based component 
c. Reference its public properties on a page 
d. Change one of its public properties 

2. Replicate the same component, but change it to use only private properties, being sure 
to include an initialization function 

a. Restrict a method by access type 
b. Restrict a method based on user role 
c. View its documentation in the same browser and editor 
d. Add a getter and setter method for one of the private properties 
e. Utilize OnMissingMethod() to make a dynamic getter/setter proxy 

3. Experiment with using the Administrator API on your local development copy of 
ColdFusion 

a. Add a datasource programmatically 
b. Add a new mapping 
c. Look at the server monitoring options 

4. Prototype a component 
a. Create an interface 
b. Create a component to validate against the interface and try validating it when 

following, and not following, the interface specifications 

